ガウスの積分公式
Gaussian quadrature

この組み合わせがよく使用される。4点におけるfの値を使用して領域における積分の値を近似する。



ガウスの積分公式
Gaussian quadrature

この組み合わせがよく使用される。4点におけるfの値を使用して領域における積分の値を近似する。



$$I=\int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\,d\xi d\eta=\sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i\cdot W_j\cdot f(\xi_i,\eta_j)\right]$$

$$=1.0\times1.0\times f(-0.57735,-0.57735)+1.0\times1.0\times f(-0.57735,+0.57735)$$

$$+1.0\times1.0\times f(+0.57735,+0.57735)+1.0\times1.0\times f(+0.57735,-0.57735)$$

宿題

- ガウスの積分公式を使用して以下の四角形の面積を求めよ



1：(1.0, 1.0)
2：(4.0, 2.0)
3：(3.0, 5.0)
4：(2.0, 4.0)

$$I = \int_V dV$$

ヒント

- 座標値によってヤコビアン（ヤコビの行列）を計算する。
- ガウスの積分公式（n=2）に代入する。

$$I=\int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\,d\xi d\eta=\sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i\cdot W_j\cdot f(\xi_i,\eta_j)\right]$$

```
implicit REAL*8 (A-H,O-Z)
real*8 W(2)
real*8 POI(2)

W(1)= 1.0d0
W(2)= 1.0d0
POI(1)= -0.5773502692d0
POI(2)= +0.5773502692d0

SUM= 0.d0
do jp= 1, 2
do ip= 1, 2
  FC = F(POI(ip),POI(jp))
  SUM= SUM + W (ip)*W (jp)*FC
enddo
enddo
```



正 解

1：(1.0, 1.0)
2：(4.0, 2.0)
3：(3.0, 5.0)
4：(2.0, 4.0)



から四隅の三角形を取り除く

$$3\times 4-\frac{1}{2}\left(3+3+2+4\right)\times 1$$

$$=12-\frac{12}{2}=6$$

何をすべきか?

- 以下の積分を求めればよい：

$$I=\int_V dV=\iint dxdy=\int_{-1}^{+1}\int_{-1}^{+1}\det\left[J\right]d\xi d\eta$$

- 定義によれば：

$$I=\int_{-1}^{+1}\int_{-1}^{+1}f(\xi,\eta)\,d\xi d\eta=\sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i\cdot W_j\cdot f(\xi_i,\eta_j)\right]$$

- つまり，det [J]= f
- ということは det [J] の積分点における値を求めれば良い！

$$\det\left[J(\xi_i,\eta_j)\right]$$

初期設定（1/4）

```
      implicit REAL*8 (A-H,O-Z)

      real*8 X(4), Y(4)
      real*8 W(2), POS(2)
      real*8 SHAPE(2,2,4)
      real*8 PNQ(2,4), PNE(2,4), DETJ(2,2)

!C
!C-- POINT data
      X(1)= 1.0
      Y(1)= 1.0

      X(2)= 4.0
      Y(2)= 2.0

      X(3)= 3.0
      Y(3)= 5.0

      X(4)= 2.0
      Y(4)= 4.0

!C
!C-- Quadrature points & weighting coef.
      W(1)= +1.0000000000d+00
      W(2)= +1.0000000000d+00

      POS(1)= -0.5773502692d+00
      POS(2)= +0.5773502692d+00
```

各節点の座標

1：(1.0, 1.0)
2：(4.0, 2.0)
3：(3.0, 5.0)
4：(2.0, 4.0)



初期設定（1/4）

```
      implicit REAL*8 (A-H,O-Z)

      real*8 X(4), Y(4)
      real*8 W(2), POS(2)
      real*8 SHAPE(2,2,4)
      real*8 PNQ(2,4), PNE(2,4), DETJ(2,2)

!C
!C-- POINT data
      X(1)= 1.0
      Y(1)= 1.0

      X(2)= 4.0
      Y(2)= 2.0

      X(3)= 3.0
      Y(3)= 5.0

      X(4)= 2.0
      Y(4)= 4.0

!C
!C-- Quadrature points & weighting coef.
      W(1)= +1.0000000000d+00
      W(2)= +1.0000000000d+00

      POS(1)= -0.5773502692d+00
      POS(2)= +0.5773502692d+00
```

POS: 積分点座標
W: 重み係数



積分点における形状関数，その微分（2/4）

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

$$QP1(i)=(1+\xi_i),\quad QM1(i)=(1-\xi_i)$$

$$EP1(j)=(1+\eta_j),\quad EM1(j)=(1-\eta_i)$$

積分点における形状関数，その微分（2/4）

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

$\boldsymbol{SHAPE}$: (ξ_i, η_j) における形状関数の値

$$N_1(\xi_i,\eta_j) = \frac{1}{4}(1-\xi_i)(1-\eta_j)$$

$$N_2(\xi_i,\eta_j) = \frac{1}{4}(1+\xi_i)(1-\eta_j)$$

$$N_3(\xi_i,\eta_j) = \frac{1}{4}(1+\xi_i)(1+\eta_j)$$

$$N_4(\xi_i,\eta_j) = \frac{1}{4}(1-\xi_i)(1+\eta_j)$$



積分点における形状関数，その微分（2/4）

```
!C
!C-- SHAPE functions
      O4th= 0.25d0

      do jp= 1, 2
      do ip= 1, 2
        QP1= 1.d0 + POS(ip)
        QM1= 1.d0 - POS(ip)
        EP1= 1.d0 + POS(jp)
        EM1= 1.d0 - POS(jp)

        SHAPE(ip,jp,1)= O4th * QM1 * EM1
        SHAPE(ip,jp,2)= O4th * QP1 * EM1
        SHAPE(ip,jp,3)= O4th * QP1 * EP1
        SHAPE(ip,jp,4)= O4th * QM1 * EP1

        PNQ(jp,1)= - O4th * EM1
        PNQ(jp,2)= + O4th * EM1
        PNQ(jp,3)= + O4th * EP1
        PNQ(jp,4)= - O4th * EP1

        PNE(ip,1)= - O4th * QM1
        PNE(ip,2)= - O4th * QP1
        PNE(ip,3)= + O4th * QP1
        PNE(ip,4)= + O4th * QM1
      enddo
      enddo
```

$$PNQ(j,k)=\frac{\partial N_k}{\partial \xi}(\xi=\xi_i,\eta=\eta_j)$$

$$PNE(j,k)=\frac{\partial N_k}{\partial \eta}(\xi=\xi_i,\eta=\eta_j)$$

$$\frac{\partial N_1}{\partial \xi}(\xi_i,\eta_j)=-\frac{1}{4}(1-\eta_j) \qquad \frac{\partial N_1}{\partial \eta}(\xi_i,\eta_j)=-\frac{1}{4}(1-\xi_i)$$

$$\frac{\partial N_2}{\partial \xi}(\xi_i,\eta_j)=+\frac{1}{4}(1-\eta_j) \qquad \frac{\partial N_2}{\partial \eta}(\xi_i,\eta_j)=-\frac{1}{4}(1+\xi_i)$$

$$\frac{\partial N_3}{\partial \xi}(\xi_i,\eta_j)=+\frac{1}{4}(1+\eta_j) \qquad \frac{\partial N_3}{\partial \eta}(\xi_i,\eta_j)=+\frac{1}{4}(1+\xi_i)$$

$$\frac{\partial N_4}{\partial \xi}(\xi_i,\eta_j)=-\frac{1}{4}(1+\eta_j) \qquad \frac{\partial N_4}{\partial \eta}(\xi_i,\eta_j)=+\frac{1}{4}(1-\xi_i)$$

(ξ_i,η_j) における形状関数の一階微分

積分点におけるヤコビアン計算（3/4）

```
!C
!C +-----------------+
!C | JACOBIAN matrix |
!C +-----------------+
!C===
      do jp= 1, 2
      do ip= 1, 2
        dXdQ = PNQ(jp,1)*X(1) + PNQ(jp,2)*X(2) +
     &         PNQ(jp,3)*X(3) + PNQ(jp,4)*X(4)
        dYdQ = PNQ(jp,1)*Y(1) + PNQ(jp,2)*Y(2) +
     &         PNQ(jp,3)*Y(3) + PNQ(jp,4)*Y(4)
        dXdE = PNE(ip,1)*X(1) + PNE(ip,2)*X(2) +
     &         PNE(ip,3)*X(3) + PNE(ip,4)*X(4)
        dYdE = PNE(ip,1)*Y(1) + PNE(ip,2)*Y(2) +
     &         PNE(ip,3)*Y(3) + PNE(ip,4)*Y(4)
        DETJ(ip,jp)= dXdQ*dYdE - dXdE*dYdQ
      enddo
      enddo
!C===
```

$$dXdQ = \frac{\partial x}{\partial \xi} \quad dYdQ = \frac{\partial y}{\partial \xi}$$

$$dXdE = \frac{\partial x}{\partial \eta} \quad dYdE = \frac{\partial y}{\partial \eta}$$

$$DETJ(i,j) = \det\left[J\left(\xi_i, \eta_j\right)\right]$$

$$J_{11} = \frac{\partial x}{\partial \xi} = \frac{\partial}{\partial \xi}\left(\sum_{i=1}^{4} N_i x_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \xi} x_i, \quad J_{12} = \frac{\partial y}{\partial \xi} = \frac{\partial}{\partial \xi}\left(\sum_{i=1}^{4} N_i y_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \xi} y_i,$$

$$J_{21} = \frac{\partial x}{\partial \eta} = \frac{\partial}{\partial \eta}\left(\sum_{i=1}^{4} N_i x_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \eta} x_i, \quad J_{22} = \frac{\partial y}{\partial \eta} = \frac{\partial}{\partial \eta}\left(\sum_{i=1}^{4} N_i y_i\right) = \sum_{i=1}^{4} \frac{\partial N_i}{\partial \eta} y_i$$

数値積分実施（4/4）

```
!C
!C +-------+
!C | AREA |
!C +-------+
!C===
      AREA= 0.d0
      do jp= 1, 2
      do ip= 1, 2
        AREA= AREA + dabs(DETJ(ip,jp)) * W(ip) * W(jp)
      enddo
      enddo

!C
!C-- ANALYTICAL SOLUTION
      XA2= X(2) - X(1)
      YA2= Y(2) - Y(1)
      XA3= X(3) - X(1)
      YA3= Y(3) - Y(1)
      XA4= X(4) - X(1)
      YA4= Y(4) - Y(1)

      AREAa= 0.50d0 * (dabs(XA2*YA3-YA2*XA3) +dabs(XA3*YA4-YA3*XA4))
!C===

      write (*,'(a,1pe16.6)') 'Gaussian quadrature', AREA
      write (*,'(a,1pe16.6)') 'analytical sol.    ', AREAa

      stop
      end
```

$$I = \int_{-1}^{+1}\int_{-1}^{+1} f(\xi,\eta)\, d\xi d\eta = \sum_{i=1}^{m}\sum_{j=1}^{n}\left[W_i \cdot W_j \cdot f(\xi_i,\eta_j)\right]$$